
*
*    R0A= 1/A0A                   R0B= 1/A0B                 R0C= 1/A0C
*   _/\  /\_                 _/\  /\__                    _/\  /\_
*   |  \/   |    R3A= 1/A0A  |  \/    |       R3B= 1/A0B |   \/   |  R3C= 1/A0C
*   |     __|_/\  /\  /\_____|_____   |_/\  /\  /\_______|_____   |_/\  /\  /\_____________
*   |     |     \/  \/          LPA|  |   \/  \/            LPB|  |   \/  \/            LPC|
*  _|_    |    R2A= 1/A1A          |  |    R2B= 1/A1B          |  |    R2C= 1/A1C          |
* /VIN\   |_/\  /\  /\_____        |  |_/\  /\  /\_____        |  |_/\  /\  /\_____        |
* \___/   |   \/  \/       |       |  |   \/  \/       |       |  |   \/  \/       |       |
*   |     |   R1A= 1       |       |  |   R1 = 1       |       |  |   R1C= 1       |       |
*  _|_    |_/\  /\_        |BPA    |  |_/\  /\_        | BPB   |  |_/\  /\_        | BPC   |
*  ///    |   \/   |       |       |  |   \/   |       |       |  |   \/   |       |       |
*         | |\  HPA|XONES1A|XONES2A|  | |\  HPB|XONES1B|XONES2B|  | |\  HPC|XONES1C|XONES2C|
*         |_|-\    | ____  | ____  |  |_|-\    | ____  | ____  |  |_|-\    | ____  | ____  |
*       SUMA|  \___|_|1/s|_|_|1/s|_|    |  \___|_|1/s|_|_|1/s|_|    |  \___|_|1/s|_|_|1/s|_|
*           |  /     |___|   |___| SUMB |  /     |___|   |___|  SUMC|  /     |___|   |___|
*         __|+/ OPAA                  __|+/ OPAB                  __|+/ OPAC
*        |  |/                       |  |/                       |  |/
*       _|_                         _|_                         _|_
*       ///                         ///                         ///
*      Bessel terms
*      (s^2 +1.861s +3.63)*(s^2 +2.76s +2.85)*(s^2 +3.143s + 2.57)
.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN    0      PWL(   0 0   2 0 2.05 1 2.1 1  2.15 0       15 0  15.5 1  30 1   30.5  0) AC = 1
R0A       VIN    SUMA   .333    
R1A       SUMA   HPA    1
R2A       SUMA   BPA    .54
R3A       SUMA   LPA    .333
XOPA1A    SUMA   0      HPA    OPA
XONES1A   HPA    BPA           ONE_S
XONES2A   BPA    LPA           ONE_S
R0B       LPA    SUMB   .35    
R1B       SUMB   HPB    1
R2B       SUMB   BPB    .362
R3B       SUMB   LPB    .35
XOPA1B    SUMB   0      HPB    OPA
XONES1B   HPB    BPB           ONE_S
XONES2B   BPB    LPB           ONE_S
R0C       LPB    SUMC   .389    
R1C       SUMC   HPC    1
R2C       SUMC   BPC    .318
R3C       SUMC   LPC    .389
XOPA1C    SUMC   0      HPC    OPA
XONES1C   HPC    BPC           ONE_S
XONES2C   BPC    LPC           ONE_S

*==A_Bessel_is_Best_for_Low_Phase_Distortion====
.control
set outfile =             "transTest.txt"
*op
tran      1     40      0 
run
plot    -1*lpc  vin
ac         dec    500     .01   10
run
plot    db(lpc) 
.endc
.SUBCKT OPA  INP    INN    OUT
*E_GAIN#    NODE_P NODE_N CNTL_P CNTL_N GAIN
*R_NUMB     NODE1  NODE2  RVALUE
*C_NUMB     NODE1  NODE2  CVALUE  
EGN1        O1     0      INP    INN    -1
EGN2        OUT    0      IP     0      -1000000
R1          O1     IP     10k 
C1          OUT    IP     10p
.ends
.SUBCKT ONE_S IN      OUT
*E_GAIN#    NODE_P   NODE_N CNTL_P CNTL_N GAIN
*R_NUMB     NODE1    NODE2  RVALUE
*C_NUMB     NODE1    NODE2  CVALUE  
EGN1        O1       0      IP     0      -1000000
EGN2        OUT      0      O1     0      -1
R1          IN       IP     1 
C1          IP       O1     1
.ends



*
*
*             C1 1F               ___
*             _||______ O1   _____|OUT|
*  ___    R1 | || _|_  |    _|_   |___|     .SUBCKT   ONE_S  IN   OUT
* |IN |_/\  _|__ /   \ |__ /   \            ___  _______    ___
* |___|   \/  IP| Egn1|   | Egn2|   ---->  |IN |_|      |__|OUT|
*       1      _| 90dB|  _|  -1 |   ---->  |___| | 1/s  |  |___|
*             |  \___/  |  \___/                 |______|
*             |____|    |____|
*                 _|_       _|_       R = 1 = 1/sC   s = 1 =  2*PI*Freq
*                 ///       ///
*
*
*    ___         O1         C1 10pF   ___     .SUBCKT  OPA INP INN OUT
*   |INP|_     _____        _||______|OUT|    
*   |___| |   _|_   |   R1 | || _|_  |___|     ___   |\
*         |_ /   \  |_/\  _|__ /   \          |INP|__|-\    ___
*    ___    | EGN1|  10K\/  IP| EGN2|  ---->  |___|  |  \__|OUT|
*   |INN|___|  -1 |          _| 90dB|  ---->   ___   |  /  |___|
*   |___|    \___/       IP |  \___/          |INN|__|+/
*              |            |____|            |___|  |/ OPA
*             _|_               _|_
*             ///               ///

.end

=====================END_OF_SPICE=======================

A Bessel is designed for Low Phase Distortion

The goal is for fundamental and all its harmonics
to all have the same input/output delay.
The group delay is therefore good  

A Bessel doesn not as Good Flat Band Response
as a Butterworth filter.




